// File: contracts/CivVault.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.17;

import "@openzeppelin/contracts/utils/math/SafeMath.sol";
import "@openzeppelin/contracts/utils/math/Math.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/token/ERC20/extensions/ERC20Burnable.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/access/AccessControl.sol";

import '@uniswap/v2-core/contracts/interfaces/IUniswapV2Pair.sol';
import '@uniswap/v2-core/contracts/interfaces/IUniswapV2Factory.sol';
import '@uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router02.sol';

import "./interfaces/ICivFund.sol";
import './libraries/FixedPoint.sol';
import './libraries/UniswapV2OracleLibrary.sol';

/// @title  Civ Vault
/// @author Ren


contract CivVault is Ownable, ReentrancyGuard {
    using SafeMath for uint256;
    using SafeERC20 for IERC20;
    using SafeERC20 for ICivFundRT;
    using FixedPoint for *;

    /// @notice Guarantee Fee Amount
    uint256 public guarantee_fee = 1000; //10% of $ value equivalent, expressed in bps
    /// @notice Fee Base Amount(Fee calculated like this guarantee_fee / feeBase)
    uint256 public constant feeBase = 10000;
    /// @notice Dead Address
    address public constant NULL_ADDRESS = 0x0000000000000000000000000000000000000000;

    struct UserInfo {
        uint256 startingEpoch;         // Starting Epoch of guarantee lock time
    }

    struct AddPoolParam {
        IERC20 _lpToken;
        ICivFundRT _fundRepresentToken;
        IERC20 _guaranteeToken;
        uint256 _maxDeposit;
        uint256 _maxUser;
        address[] _withdrawAddresses; // Withdraw Address
        uint256 _fee;
        uint256 _feeDuration;
        uint256 _depositDuration;
        uint256 _withdrawDuration;   
        uint256 _lockPeriod; 
        bool _paused;
    }

    struct DepositParams {
        uint256 depositInfo;
        uint256 withdrawInfo;
        uint256 depositQuantity;
    }

    /// @notice vault getter contract
    ICivVaultGetter public vaultGetter;
    /// @notice structure with info on each pool
    PoolInfo[] public poolInfo;
    /// @notice structure with info on each pool
    VaultInfo[] public vaultInfo;
    /// @notice Info of each user that enters the fund
    mapping(uint256 => mapping(address => UserInfo)) public userInfo; 
    /// @notice Info if represent token is already added to the pool
    mapping(address=> bool) public bFundRepresentTokenAdded;
    /// @notice Each Pools deposit informations
    mapping(uint256 => mapping( address=> mapping(uint256 => DepositParams))) public depositParams;
    /// @notice Each Pools deposit time values
    mapping(uint256 => mapping(uint256 => uint256)) public depositTime;

    /// @notice Event emitted when user deposit fund to our vault or vault deposit fund to strategy
    event Deposit(address indexed user, address receiver, uint256 indexed pid, uint256 amount);
    /// @notice Event emitted when user request withdraw fund from our vault or vault withdraw fund to user
    event Withdraw(address indexed user, uint256 indexed pid, uint256 amount);
    /// @notice Event emitted when owner sets new fee
    event SetFee(uint256 pid, uint256 oldFee, uint256 newFee);
    /// @notice Event emitted when owner sets new guarantee fee
    event SetGuaranteeFee(uint256 oldFee, uint256 newFee);
    /// @notice Event emitted when owner sets new fee duration
    event SetFeeDuration(uint256 pid, uint256 oldDuration, uint256 newDuration);
    /// @notice Event emitted when owner sets new deposit duration
    event SetDepositDuration(uint256 pid, uint256 oldDuration, uint256 newDuration);
    /// @notice Event emitted when owner sets new guarantee token lock time
    event SetPoolLockTime(uint256 pid, uint256 oldLocktime, uint256 newLockTime);
    /// @notice Event emitted when owner sets new withdraw duration
    event SetWithdrawDuration(uint256 pid, uint256 oldDuration, uint256 newDuration);
    /// @notice Event emitted when owner sets new treasury addresses
    event SetWithdrawAddress(uint256 pid, address[] oldAddress, address[] newAddress);
    /// @notice Event emitted when send fee to our treasury
    event SendFeeWithOwner(uint256 pid, address treasuryAddress,uint256 feeAmount);
    /// @notice Event emitted when owner update new NAV
    event UpdateNAV(uint256 pid, uint256 NAV, uint256 watermark, uint256 unpaidFee);
    /// @notice Event emitted when owner paused deposit
    event SetPaused(uint256 pid, bool paused);
    /// @notice Event emitted when owner set new Max Deposit Amount
    event SetMaxDeposit(uint256 pid, uint256 oldMaxAmount, uint256 newMaxAmount);
    /// @notice Event emitted when owner set new Max User Count for Deposit/Withdraw
    event SetMaxUser(uint256 _pid, uint256 oldMaxUser, uint256 newMaxUser);
    /// @notice Event emitted when user withdraw pending amount from vault
    event WithdrawPending(address user,uint256 pid, uint256 amount);
    /// @notice Event emitted when Uniswap Token Price Updated
    event Update(uint256 _pid, uint256 index);
    /// @notice Event emitted when user claim guarantee token
    event ClaimGuarantee(uint256 pid, address user, uint256 rewardAmount);
    /// @notice Event emitted when user claim LP token for each epoch
    event ClaimWithdrawedToken(uint256 pid, address user, uint256 epoch, uint256 lpAmount);
    /// @notice Event emitted when user claim LP token
    event GetWithdrawedToken(uint256 pid, address user, uint256 lpAmount);
    /// @notice Event emitted when owner adds new pool
    event AddPool(
        uint256 indexed pid,
        uint256 indexed fee,
        uint256 maxDeposit,
        bool paused,
        address[] withdrawAddress,
        address lpToken,
        address guaranteeToken,
        uint256 lockPeriod
    );

    modifier whenDepositPaused(uint256 _pid) {
        require(vaultInfo.length > _pid, "Pool does not exist");
        require(vaultInfo[_pid].paused == true, "Deposit is not paused");
        _;
    }
    modifier whenDepositNotPaused(uint256 _pid) {
        require(vaultInfo.length > _pid, "Pool does not exist");
        require(vaultInfo[_pid].paused == false, "deposit paused");
        _;
    }

    constructor(
    ) {}

    /// @notice Add new pool to our vault
    /// @dev Only Owner can call this function
    /// @param addPoolParam Parameters for new pool
    function addPool(
        AddPoolParam memory addPoolParam
    ) public virtual onlyOwner {
        require(addPoolParam._withdrawAddresses.length == 2,"Treasury Address Length must be 2");
        require(address(addPoolParam._fundRepresentToken) != NULL_ADDRESS,"Fund Represent Token address cannot be null address");
        require(address(addPoolParam._guaranteeToken) != NULL_ADDRESS,"Guarantee Token address cannot be null address");
        require(addPoolParam._withdrawAddresses[0] != NULL_ADDRESS,"first Treasury address cannot be null address");
        require(addPoolParam._withdrawAddresses[1] != NULL_ADDRESS,"second Treasury address cannot be null address");
        require(!bFundRepresentTokenAdded[address(addPoolParam._fundRepresentToken)], "represent token already added");
        poolInfo.push(
            PoolInfo({
                lpToken: addPoolParam._lpToken, // Rewardable contract: token for staking, LP for Funding, or NFT for NFT staking
                fundRepresentToken: addPoolParam._fundRepresentToken,
                guaranteeToken: addPoolParam._guaranteeToken,
                NAV: 0, //init = always 0
                totalShares: 0, //init = always 0
                fee: addPoolParam._fee,
                withdrawAddress: addPoolParam._withdrawAddresses,
                unpaidFee: 0,
                watermark: 0,
                burnShareAmount: 0,
                currentDepositEpoch: 0,
                currentWithdrawEpoch: 0,
                valuePerShare: 0 //init = always 0
            })
        );  
        vaultInfo.push(
            VaultInfo({
                maxDeposit: addPoolParam._maxDeposit,
                maxUser: addPoolParam._maxUser,
                paused: addPoolParam._paused,
                currentDeposit: 0,
                currentWithdraw: 0,
                collectFeeDuration: addPoolParam._feeDuration,
                lastCollectFee: block.timestamp,
                depositDuration: addPoolParam._depositDuration,
                withdrawDuration: addPoolParam._withdrawDuration,
                lockPeriod: addPoolParam._lockPeriod,
                lastDeposit: block.timestamp,
                lastWithdraw: block.timestamp,
                curDepositUser: 0,
                curWithdrawUser: 0
            })
        );
        uint256 pid = poolInfo.length - 1;
        vaultGetter.addUniPair(pid, address(addPoolParam._lpToken), address(addPoolParam._guaranteeToken));
        bFundRepresentTokenAdded[address(addPoolParam._fundRepresentToken)] = true;
        emit AddPool(
            pid,
            addPoolParam._fee,
            addPoolParam._maxDeposit,
            addPoolParam._paused,
            addPoolParam._withdrawAddresses,
            address(addPoolParam._lpToken),
            address(addPoolParam._guaranteeToken),
            addPoolParam._lockPeriod
        );     
    }

    /// @notice Sets new fee
    /// @dev Only Owner can call this function
    /// @param _pid Pool Id
    /// @param _newFee New Fee Percent
    function setFee(uint256 _pid,uint256 _newFee) external onlyOwner {
        require(poolInfo.length > _pid, "Pool does not exist");
        emit SetFee(_pid, poolInfo[_pid].fee, _newFee);
        poolInfo[_pid].fee = _newFee;
    }

    /// @notice Sets new collecting fee duration
    /// @dev Only Owner can call this function
    /// @param _pid Pool Id
    /// @param _newDuration New Collecting Fee Duration
    function setFeeDuration(uint256 _pid, uint256 _newDuration) external onlyOwner {
        require(vaultInfo.length > _pid, "Pool does not exist");
        emit SetFeeDuration(_pid, vaultInfo[_pid].collectFeeDuration, _newDuration);
        vaultInfo[_pid].collectFeeDuration = _newDuration;
    }

    /// @notice Sets new Pool guarantee token lock time
    /// @dev Only Owner can call this function
    /// @param _pid Pool Id
    /// @param _lockTime New Guarantee token lock time
    function setPoolLockTime(uint256 _pid, uint256 _lockTime)
        external
        onlyOwner
    {
        require(vaultInfo.length > _pid, "Pool does not exist");
        uint256 previousLockTime = vaultInfo[_pid].lockPeriod;
        emit SetPoolLockTime(_pid, previousLockTime, _lockTime);
        vaultInfo[_pid].lockPeriod = _lockTime;
    }

    /// @notice Sets new deposit fund from vault to strategy duration
    /// @dev Only Owner can call this function
    /// @param _pid Pool Id
    /// @param _newDuration New Duration for Deposit fund from vault to strategy
    function setDepositDuration(uint256 _pid, uint256 _newDuration) external onlyOwner {
        require(vaultInfo.length > _pid, "Pool does not exist");
        emit SetDepositDuration(_pid, vaultInfo[_pid].depositDuration, _newDuration);
        vaultInfo[_pid].depositDuration = _newDuration;
    }

    /// @notice Sets new withdraw fund from strategy to users duration
    /// @dev Only Owner can call this function
    /// @param _pid Pool Id
    /// @param _newDuration New Duration for Withdraw fund from strategy to users
    function setWithdrawDuration(uint256 _pid, uint256 _newDuration) external onlyOwner {
        require(vaultInfo.length > _pid, "Pool does not exist");
        emit SetWithdrawDuration(_pid, vaultInfo[_pid].withdrawDuration, _newDuration);
        vaultInfo[_pid].withdrawDuration = _newDuration;
    }

    /// @notice Sets new treasury addresses to keep fee
    /// @dev Only Owner can call this function
    /// @param _pid Pool Id
    /// @param _newAddress Address list to keep fee
    function setWithdrawAddress(uint256 _pid, address[] memory _newAddress) external onlyOwner {
        require(_newAddress.length == 2, "Withdraw Addresses length must be 2");
        require(_newAddress[0] != NULL_ADDRESS,"first Treasury address cannot be null address");
        require(_newAddress[1] != NULL_ADDRESS,"second Treasury address cannot be null address");
        require(poolInfo.length > _pid, "Pool does not exist");
        emit SetWithdrawAddress(_pid, poolInfo[_pid].withdrawAddress, _newAddress);
        poolInfo[_pid].withdrawAddress = _newAddress;
    }

    /// @notice Set Pause of Unpause for deposit to vault
    /// @dev Only Owner can change this status
    /// @param _pid Pool Id
    /// @param _paused paused or unpaused for deposit
    function setPaused(uint256 _pid,bool _paused) external onlyOwner {
        require(vaultInfo.length > _pid, "Pool does not exist");
        emit SetPaused(_pid, _paused);
        vaultInfo[_pid].paused = _paused;
    }

    /// @notice Set Max Deposit Amount in the vault on a given pool
    /// @dev Only Owner can change this status
    /// @param _pid Pool Id
    /// @param _newMaxDeposit New Max Deposit Amount
    function setMaxDeposit(uint256 _pid,uint256 _newMaxDeposit) external onlyOwner {
        require(vaultInfo.length > _pid, "Pool does not exist");
        emit SetMaxDeposit(_pid, vaultInfo[_pid].maxDeposit, _newMaxDeposit);
        vaultInfo[_pid].maxDeposit = _newMaxDeposit;
    }

    /// @notice Set Max Deposit/Withdraw User Count in the vault on a given pool
    /// @dev Only Owner can change this status
    /// @param _pid Pool Id
    /// @param _newMaxUser New Max User Count
    function setMaxUser(uint256 _pid,uint256 _newMaxUser) external onlyOwner {
        require(vaultInfo.length > _pid, "Pool does not exist");
        emit SetMaxUser(_pid, vaultInfo[_pid].maxUser, _newMaxUser);
        vaultInfo[_pid].maxUser = _newMaxUser;
    }

    /// @notice Sets new guarantee fee
    /// @dev Only Owner can call this function
    /// @param _newFee new guarantee fee amount
    function setGuaranteeFee(uint256 _newFee) external onlyOwner {
        emit SetGuaranteeFee(guarantee_fee, _newFee);
        guarantee_fee = _newFee;
    }

    /// @notice Sets vault getter function
    /// @dev Only Owner can call this function
    /// @param _vaultGetter vaultGetter function address
    function setVaultGetter(ICivVaultGetter _vaultGetter) external onlyOwner {
        vaultGetter = _vaultGetter;
    }

    /// @notice Sets new NAV of the pool.
    /** 
     * @dev Only Owner can call this function. 
     *      Owner must transfer fund to our vault before calling this function
    */ 
    /// @param _pid Pool Id
    /// @param _newNAV New NAV value
    function updateNAV(uint256 _pid, uint256 _newNAV) external onlyOwner whenDepositPaused(_pid) {
        PoolInfo storage pool = poolInfo[_pid];
        pool.NAV = _newNAV;
        if( pool.watermark < _newNAV) {
            uint256 actualFee = (_newNAV - pool.watermark) * pool.fee / feeBase;
            if( actualFee > 0 ) {
                pool.watermark = _newNAV - actualFee;
                pool.unpaidFee += actualFee;
            }
        }
        emit UpdateNAV(_pid, _newNAV, pool.watermark, pool.unpaidFee);
    }

    function getPoolInfo(uint256 _pid) external view returns(PoolInfo memory pool, VaultInfo memory vault) {
        pool = poolInfo[_pid];
        vault = vaultInfo[_pid];
    }

    /// @dev Number of pools
    /// @return length Current Pool Length in a smart contract
    function poolLength() external view returns (uint256) {
        return poolInfo.length;
    }

    /// @dev Get Guarantee amount for deposit to the vault
    /// @param _pid Pool Id
    /// @param amount Amount to deposit in the vault
    /// @return amount Guarantee Token Amount needs for deposit in a given pool
    function getDepositGuarantee(uint256 _pid,uint256 amount) external view returns(uint) {
        return vaultGetter.getPrice(_pid, amount) * guarantee_fee / feeBase;
    }

    /// @dev Get Current Epoch deposit info on a given pool
    /// @param _pid Pool Id
    /// @param _user userAddress
    /// @return _curEpoch epoch value
    function getDepositParams(uint256 _pid,address _user, uint256 _curEpoch) external view returns(DepositParams memory) {
        return depositParams[_pid][_user][_curEpoch];
    }

    /// @dev Get Guarantee Token symbol and decimal
    /// @param _pid Pool Id
    /// @return symbol Guarantee Token Symbol in a given pool
    /// @return decimals Guarantee Token Decimal in a given pool
    function getGuaranteeTokenInfo(uint256 _pid) external view returns(string memory symbol, uint decimals) {
        IERC20Extended guarantee = IERC20Extended(address(poolInfo[_pid].guaranteeToken));
        symbol = guarantee.symbol();
        decimals = guarantee.decimals();
    }

    /// @dev Get available deposit amount based of user's guarantee amount
    /// @param _pid Pool Id
    /// @return amount Current Available Deposit amount regarding users's current guarantee token balance in a given pool
    function getAvailableDeposit(uint256 _pid) external view returns(uint) {
        IERC20Extended guarantee = IERC20Extended(address(poolInfo[_pid].guaranteeToken));
        uint256 balance = guarantee.balanceOf(_msgSender());
        return vaultGetter.getReversePrice(_pid, balance) * feeBase / guarantee_fee;
    }

    /// @dev Get claimable guarantee token amount
    /// @param _pid Pool Id
    /// @param _user userAddress
    /// @return amount Current claimable guarantee token amount
    function getClaimableGuaranteeToken(uint256 _pid, address _user)
        external
        view
        returns (uint256)
    {
        VaultInfo memory vault = vaultInfo[_pid];
        UserInfo storage user = userInfo[_pid][_user];
        uint256 curEpoch = poolInfo[_pid].currentDepositEpoch;
        uint256 unLocked;
        for( uint256 i = user.startingEpoch; i < curEpoch; i++ ) {
            if (
                block.timestamp >= depositTime[_pid][i] + vault.lockPeriod
            ) {
                unLocked += depositParams[_pid][_user][i].depositQuantity;
            }
        }

        return unLocked;
    }

    /// @notice Get withdraw tokens from vault
    /**
     * @dev Withdraw my fund from vault
     */
    /// @param _pid Pool Id
    function getWithdrawedToken(uint256 _pid, uint256[] memory epochs) external nonReentrant whenDepositNotPaused(_pid) {
        require(poolInfo.length > _pid, "Pool does not exist");
        PoolInfo memory pool = poolInfo[_pid];
        uint256 unclaimedAmount = 0;
       
        for(uint i = 0; i < epochs.length;i++) {
            if( epochs[i] < pool.currentWithdrawEpoch) {
                DepositParams storage dp = depositParams[_pid][_msgSender()][epochs[i]];
                unclaimedAmount += dp.withdrawInfo;
                emit ClaimWithdrawedToken(_pid, _msgSender(), epochs[i], dp.withdrawInfo);
                dp.withdrawInfo = 0;
            }
        }
        pool.lpToken.safeTransfer(_msgSender(), unclaimedAmount);
        emit GetWithdrawedToken(_pid, _msgSender(), unclaimedAmount);
    }

    /// @notice get unclaimed withdrawed token epochs
    /// @param _pid Pool Id
    /// @return _epochs array of unclaimed epochs
    function getUnclaimedTokenEpochs(uint256 _pid, address user) external view returns(uint256[] memory _epochs, uint256 _epochLength) {
        require(poolInfo.length > _pid, "Pool does not exist");
        PoolInfo memory pool = poolInfo[_pid];
        uint256 epochLen = 0;
        uint256 curWithdrawEpoch = pool.currentWithdrawEpoch;
        uint256[] memory tempEpochs = new uint256[](curWithdrawEpoch);
        for(uint i = 0; i < curWithdrawEpoch; i++) {
            if( depositParams[_pid][user][i].withdrawInfo > 0 ) {
                tempEpochs[epochLen++] = i;
            }
        }
        _epochLength = epochLen;
        _epochs = new uint256[](_epochLength);
        for(uint i = 0; i < _epochLength; i++) {
            _epochs[i] = tempEpochs[i];
        }
    }

    /// @notice Claim withdrawed token epochs
    /// @param _pid Pool Id
    /// @param _maxEpoch max epoch count to claim  
    function claimGuaranteeToken(uint256 _pid, uint256 _maxEpoch) external {
        require(poolInfo.length > _pid, "Pool does not exist");
        VaultInfo memory vault = vaultInfo[_pid];
        PoolInfo memory pool = poolInfo[_pid];
        UserInfo storage user = userInfo[_pid][_msgSender()];
        uint256 curEpoch =  pool.currentDepositEpoch;
        uint256 endEpoch = Math.min(_maxEpoch, curEpoch);
        uint256 _startingEpoch = user.startingEpoch;
        uint256 _startingEpochFinal = _startingEpoch;
        uint256 actualReward;
        for( uint256 i = user.startingEpoch; i < endEpoch; i++ ) {
            if (block.timestamp < depositTime[_pid][i] + vault.lockPeriod) {
                break;
            }
            actualReward += depositParams[_pid][_msgSender()][i].depositQuantity;
            _startingEpochFinal = i + 1;
        }
        require(actualReward > 0, "no reward");
        user.startingEpoch = _startingEpochFinal;
        pool.guaranteeToken.safeTransfer(_msgSender(), actualReward);
        emit ClaimGuarantee(_pid, _msgSender(), actualReward);
    }

    /// @notice Users Deposit tokens to our vault
    /**
     * @dev Anyone can call this function if pool is not paused.
     *      Users must approve deposit token before calling this function
     *      We mint represent token to users so that we can calculate each users deposit amount outside
     */ 
    /// @param _pid Pool Id
    /// @param _amount Token Amount to deposit
    function deposit(uint256 _pid, uint256 _amount)
        external
        nonReentrant
        whenDepositNotPaused(_pid)
    {
        require(poolInfo.length > _pid, "Pool does not exist");
        require(vaultInfo[_pid].paused==false,"deposit paused");
        PoolInfo storage pool = poolInfo[_pid];
        VaultInfo storage vault = vaultInfo[_pid];
        uint256 curEpoch = pool.currentDepositEpoch;
        DepositParams storage dp = depositParams[_pid][_msgSender()][curEpoch];
        require( vault.currentDeposit+_amount <= vault.maxDeposit, "exceeds limit");
        if(dp.depositInfo == 0) {
            require(vault.curDepositUser < vault.maxUser, "Can't deposit any more");
            vault.curDepositUser++;
        }
        if (_amount > 0) { 
            if( pool.unpaidFee > 0 ) {
                sendFee(_pid, pool.unpaidFee);
            }
            // transfer guarantee token to the vault
            vaultGetter.updateAll(_pid);
            
            uint256 guaranteeAmount = vaultGetter.getPrice(_pid,_amount) * guarantee_fee / feeBase;
            require(guaranteeAmount>0, "not supported");
            pool.guaranteeToken.safeTransferFrom(_msgSender(), address(this), guaranteeAmount);
            dp.depositQuantity += guaranteeAmount;
            
            if( pool.totalShares == 0 ) {
                pool.valuePerShare = _amount;
            } else {
                pool.valuePerShare = pool.NAV * 10 **18 / pool.totalShares;
            }
            pool.NAV += _amount;
            uint256 addShare = _amount * 10 ** 18 / pool.valuePerShare;
            pool.totalShares += addShare;
            pool.watermark += _amount;
            pool.lpToken.safeTransferFrom(_msgSender(), address(this), _amount);
           
            if (addShare > 0) {
                pool.fundRepresentToken.mint(_msgSender(), addShare); // decimals of rewardable and represent tokens should be the same
            }
            dp.depositInfo += _amount;
            vault.currentDeposit += _amount;
        }
        emit Deposit(_msgSender(), address(this), _pid, _amount);
    }

    /// @notice Deposit vault fund to strategy address
    /**
     * @dev Only Owner can call this function if deposit duration is passed.
     *      Owner must setPaused(false)
     */
    /// @param _pid Pool Id
    /// @param invest strategy address
    function depositToFund(uint256 _pid, address invest)
        external
        nonReentrant
        whenDepositPaused(_pid)
        onlyOwner
    {
        PoolInfo storage pool = poolInfo[_pid];
        VaultInfo storage vault = vaultInfo[_pid];
        uint256 currentDeposit = vault.currentDeposit;
        require(invest != NULL_ADDRESS, "strategy address cannot be null address");
        require(currentDeposit > 0, "no token for deposit to fund");
        require(vault.lastDeposit + vault.depositDuration <= block.timestamp, "not ready for deposit");
        require(pool.lpToken.balanceOf(address(this)) >= currentDeposit);
        if( pool.unpaidFee > 0 ) {
            sendFee(_pid, pool.unpaidFee);
        }
        pool.valuePerShare = pool.NAV * 10 **18 / pool.totalShares;
        pool.lpToken.safeTransfer(invest, currentDeposit);
        depositTime[_pid][pool.currentDepositEpoch] = block.timestamp;
        pool.currentDepositEpoch++;
        vault.curDepositUser = 0;
        vault.currentDeposit = 0;
        vault.lastDeposit = block.timestamp;
        emit Deposit(address(this), invest, _pid, currentDeposit);
    }

    /// @notice Sends Withdraw Request to vault
    /**
     * @dev Withdraw all users fund from vault
     */
    /// @param _pid Pool Id
    function withdrawAll(uint256 _pid) external nonReentrant whenDepositNotPaused(_pid) {
        require(poolInfo.length > _pid, "Pool does not exist");
        VaultInfo storage vault = vaultInfo[_pid];
        PoolInfo storage pool = poolInfo[_pid];
        if( pool.unpaidFee > 0 ) {
            sendFee(_pid, pool.unpaidFee);
        }
        (uint256 currentBalance,uint256 shareAmount) = vaultGetter.getBalanceOfUser(_pid,_msgSender());
        pool.valuePerShare = pool.NAV * 10 **18 / pool.totalShares;
        uint256 valuePerShare = pool.valuePerShare;


        uint256 curDepositEpoch = pool.currentDepositEpoch;
        DepositParams storage depositParam = depositParams[_pid][_msgSender()][curDepositEpoch];
        require(currentBalance > 0, "Current balance must be bigger than 0");
        require( currentBalance > depositParam.depositInfo, "rt balance is less than deposit amount");
        currentBalance -= depositParam.depositInfo;
        uint256 curWithdrawEpoch = pool.currentWithdrawEpoch;
        DepositParams storage dp = depositParams[_pid][_msgSender()][curWithdrawEpoch];
        if(dp.withdrawInfo == 0) {
            require(vault.curWithdrawUser < vault.maxUser, "Can't withdraw any more");
            vault.curWithdrawUser++;
        }
        uint256 shareTransferAmount = currentBalance * 10 ** 18 / valuePerShare;
        require(shareAmount >= shareTransferAmount, "rt balance is less than burn amount");
        
        vault.currentWithdraw += currentBalance;
        dp.withdrawInfo += currentBalance;
        pool.fundRepresentToken.safeTransferFrom(_msgSender(),address(this), shareTransferAmount);
        pool.burnShareAmount += shareTransferAmount;
        emit Withdraw(_msgSender(), _pid, currentBalance);
    }

    /// @notice Immediately withdraw current pending deposit amount
    /// @param _pid Pool Id
    function withdrawPending(uint256 _pid) external nonReentrant whenDepositNotPaused(_pid) {
        require(poolInfo.length > _pid, "Pool does not exist");
        PoolInfo storage pool = poolInfo[_pid];
        VaultInfo storage vault = vaultInfo[_pid];
        uint256 curEpoch = pool.currentDepositEpoch;
        DepositParams storage dp = depositParams[_pid][_msgSender()][curEpoch];
        uint256 amount = dp.depositInfo;
        require( amount > 0, "no amount to withdraw pending token");
        if( pool.unpaidFee > 0 ) {
            sendFee(_pid, pool.unpaidFee);
        }
        pool.valuePerShare = pool.NAV * 10 **18 / pool.totalShares;
        uint256 valuePerShare = pool.valuePerShare;
        uint256 shareAmount = amount * 10 ** 18 / valuePerShare;
        pool.fundRepresentToken.burnFrom(_msgSender(),shareAmount);
        pool.NAV -= amount;
        pool.totalShares -= shareAmount;
        pool.watermark -= amount;
        pool.lpToken.safeTransfer(_msgSender(), amount);
        uint256 guaranteeAmount = dp.depositQuantity;
        dp.depositQuantity = 0;
        pool.guaranteeToken.safeTransfer(_msgSender(), guaranteeAmount);
        dp.depositInfo -= amount;
        vault.currentDeposit -= amount;
        vault.curDepositUser--;
        emit WithdrawPending(_msgSender(), _pid, amount);
    }

    /// @notice Sends tokens to users that request withdraw
    /**
     * @dev Owner must deposit token before calling this function
            Only Owner can call this function
            We burn fund represent token based withdraw request amount
            We send fee to the treasury address
     */
    /// @param _pid Pool Id
    function withdrawFromVault(uint256 _pid)
        external
        nonReentrant
        whenDepositPaused(_pid)
        onlyOwner
    {
        require(poolInfo.length > _pid, "Pool does not exist");
        PoolInfo storage pool = poolInfo[_pid];
        VaultInfo storage vault = vaultInfo[_pid];
        uint256 currentWithdraw = vault.currentWithdraw;
        uint256 burnShareAmount = pool.burnShareAmount;
        require( vault.curWithdrawUser > 0, "no user for withdraw");
        require(pool.totalShares > 0, "nothing to withdraw");
        require(vault.lastWithdraw + vault.withdrawDuration <= block.timestamp, "not ready for withdraw");
        require(pool.lpToken.balanceOf(address(this)) >= currentWithdraw + pool.unpaidFee, "not enough amount to withdraw");
        if( pool.unpaidFee > 0 ) {
            sendFee(_pid, pool.unpaidFee);
        }
        pool.NAV -= currentWithdraw;
        pool.watermark -= currentWithdraw;
        pool.totalShares -= burnShareAmount;
        vault.currentWithdraw = 0;
        vault.lastWithdraw = block.timestamp;
        vault.curWithdrawUser = 0;
        pool.fundRepresentToken.burn(burnShareAmount);
        pool.burnShareAmount = 0;
        pool.currentWithdrawEpoch++;
        emit Withdraw(address(this), _pid, currentWithdraw);
    }

    /// @notice Sends fee to the treasury address
    /**
     * @dev Internal function
     */
    /// @param _pid Pool Id
    /// @param feeAmount feeAmount
    function sendFee(uint256 _pid, uint256 feeAmount) internal {
        PoolInfo storage pool = poolInfo[_pid];
        pool.NAV -= feeAmount;
        address addr0 = pool.withdrawAddress[0];
        address addr1 = pool.withdrawAddress[1];
        emit SendFeeWithOwner(_pid,addr0, feeAmount / 2);
        emit SendFeeWithOwner(_pid,addr1, feeAmount / 2);
        pool.unpaidFee = 0;
        pool.lpToken.safeTransfer(addr0, feeAmount / 2);
        pool.lpToken.safeTransfer(addr1, feeAmount / 2);
    }

    /// @notice Collects protocol fee
    /**
     * @dev Only Owner can call this function
     */
    /// @param _pid Pool Id
    function collectFee(uint256 _pid) external onlyOwner {
        require(poolInfo.length > _pid, "Pool does not exist");
        PoolInfo memory pool = poolInfo[_pid];
        VaultInfo storage vault = vaultInfo[_pid];
        require(pool.unpaidFee > 0, "no fee to collect");
        require(vault.lastCollectFee + vault.collectFeeDuration <= block.timestamp,"not ready yet!");
        vault.lastCollectFee = block.timestamp;
        sendFee(_pid, poolInfo[_pid].unpaidFee);
    }

    /* Just in case anyone sends tokens by accident to this contract */

    /// @notice Transfers ETH to the recipient address
    /// @dev Fails with `STE`
    /// @param to The destination of the transfer
    /// @param value The value to be transferred
    function safeTransferETH(address to, uint256 value) internal {
        (bool success, ) = to.call{value: value}(new bytes(0));
        require(success, "CivFund");
    }

    /// @notice Withdraw ETH to the owner
    /**
     * @dev Only Owner can call this function
     */
    function withdrawETH() external payable onlyOwner {
        safeTransferETH(_msgSender(), address(this).balance);
    }

    /// @notice Withdraw ERC-20 Token to the owner
    /**
     * @dev Only Owner can call this function
     */
    /// @param _tokenContract ERC-20 Token address
    function withdrawERC20(IERC20 _tokenContract) external onlyOwner {
        for( uint256 i = 0; i < poolInfo.length; i++) {
            require(poolInfo[i].guaranteeToken != _tokenContract, "Withdraw Token cannot be deposit token");
            require(poolInfo[i].lpToken != _tokenContract, "Withdraw Token cannot be deposit token");
        }
        _tokenContract.safeTransfer(
            _msgSender(),
            _tokenContract.balanceOf(address(this))
        );
    }

    /**
     * @dev allow the contract to receive ETH
     * without payable fallback and receive, it would fail
     */
    fallback() external payable {}

    receive() external payable {}
}

// File: @openzeppelin/contracts/utils/math/SafeMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (utils/math/SafeMath.sol)

pragma solidity ^0.8.0;

// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}

// File: @openzeppelin/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a >= b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a / b + (a % b == 0 ? 0 : 1);
    }
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/ERC20Burnable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC20/extensions/ERC20Burnable.sol)

pragma solidity ^0.8.0;

import "../ERC20.sol";
import "../../../utils/Context.sol";

/**
 * @dev Extension of {ERC20} that allows token holders to destroy both their own
 * tokens and those that they have an allowance for, in a way that can be
 * recognized off-chain (via event analysis).
 */
abstract contract ERC20Burnable is Context, ERC20 {
    /**
     * @dev Destroys `amount` tokens from the caller.
     *
     * See {ERC20-_burn}.
     */
    function burn(uint256 amount) public virtual {
        _burn(_msgSender(), amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, deducting from the caller's
     * allowance.
     *
     * See {ERC20-_burn} and {ERC20-allowance}.
     *
     * Requirements:
     *
     * - the caller must have allowance for ``accounts``'s tokens of at least
     * `amount`.
     */
    function burnFrom(address account, uint256 amount) public virtual {
        _spendAllowance(account, _msgSender(), amount);
        _burn(account, amount);
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/access/AccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (access/AccessControl.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";
import "../utils/Context.sol";
import "../utils/Strings.sol";
import "../utils/introspection/ERC165.sol";

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role);
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `_msgSender()` is missing `role`.
     * Overriding this function changes the behavior of the {onlyRole} modifier.
     *
     * Format of the revert message is described in {_checkRole}.
     *
     * _Available since v4.6._
     */
    function _checkRole(bytes32 role) internal view virtual {
        _checkRole(role, _msgSender());
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(uint160(account), 20),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

// File: @uniswap/v2-core/contracts/interfaces/IUniswapV2Pair.sol



pragma solidity >=0.5.0;

interface IUniswapV2Pair {
    event Approval(address indexed owner, address indexed spender, uint value);
    event Transfer(address indexed from, address indexed to, uint value);

    function name() external pure returns (string memory);
    function symbol() external pure returns (string memory);
    function decimals() external pure returns (uint8);
    function totalSupply() external view returns (uint);
    function balanceOf(address owner) external view returns (uint);
    function allowance(address owner, address spender) external view returns (uint);

    function approve(address spender, uint value) external returns (bool);
    function transfer(address to, uint value) external returns (bool);
    function transferFrom(address from, address to, uint value) external returns (bool);

    function DOMAIN_SEPARATOR() external view returns (bytes32);
    function PERMIT_TYPEHASH() external pure returns (bytes32);
    function nonces(address owner) external view returns (uint);

    function permit(address owner, address spender, uint value, uint deadline, uint8 v, bytes32 r, bytes32 s) external;

    event Mint(address indexed sender, uint amount0, uint amount1);
    event Burn(address indexed sender, uint amount0, uint amount1, address indexed to);
    event Swap(
        address indexed sender,
        uint amount0In,
        uint amount1In,
        uint amount0Out,
        uint amount1Out,
        address indexed to
    );
    event Sync(uint112 reserve0, uint112 reserve1);

    function MINIMUM_LIQUIDITY() external pure returns (uint);
    function factory() external view returns (address);
    function token0() external view returns (address);
    function token1() external view returns (address);
    function getReserves() external view returns (uint112 reserve0, uint112 reserve1, uint32 blockTimestampLast);
    function price0CumulativeLast() external view returns (uint);
    function price1CumulativeLast() external view returns (uint);
    function kLast() external view returns (uint);

    function mint(address to) external returns (uint liquidity);
    function burn(address to) external returns (uint amount0, uint amount1);
    function swap(uint amount0Out, uint amount1Out, address to, bytes calldata data) external;
    function skim(address to) external;
    function sync() external;

    function initialize(address, address) external;
}

// File: @uniswap/v2-core/contracts/interfaces/IUniswapV2Factory.sol



pragma solidity >=0.5.0;

interface IUniswapV2Factory {
    event PairCreated(address indexed token0, address indexed token1, address pair, uint);

    function feeTo() external view returns (address);
    function feeToSetter() external view returns (address);

    function getPair(address tokenA, address tokenB) external view returns (address pair);
    function allPairs(uint) external view returns (address pair);
    function allPairsLength() external view returns (uint);

    function createPair(address tokenA, address tokenB) external returns (address pair);

    function setFeeTo(address) external;
    function setFeeToSetter(address) external;
}

// File: @uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router02.sol



pragma solidity >=0.6.2;

import './IUniswapV2Router01.sol';

interface IUniswapV2Router02 is IUniswapV2Router01 {
    function removeLiquidityETHSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountETH);
    function removeLiquidityETHWithPermitSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountETH);

    function swapExactTokensForTokensSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
    function swapExactETHForTokensSupportingFeeOnTransferTokens(
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external payable;
    function swapExactTokensForETHSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
}

// File: contracts/interfaces/ICivFund.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.17;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/access/IAccessControl.sol";
import "@openzeppelin/contracts/token/ERC20/extensions/ERC20Burnable.sol";
import '@uniswap/v2-core/contracts/interfaces/IUniswapV2Pair.sol';
import '../libraries/FixedPoint.sol';

struct PoolInfo {
    // Info on each pool
    IERC20 lpToken; // Address of asset token (Staking Token) e.g. USDT
    ICivFundRT fundRepresentToken; // Fund Represent tokens for deposit in the strategy
    IERC20 guaranteeToken;  // Guarantee Token address e.g. xStone
    uint256 NAV;            // Pool NAV
    uint256 totalShares;    // Pool Total Share Amount
    uint256 valuePerShare;  // Share Value
    uint256 watermark;      // Pool WaterMark
    uint256 unpaidFee;      // Pool Unpaid Amount
    uint256 fee;            // Pool Fee Amount
    address[] withdrawAddress;  // Pool Withdraw Address
    uint256 burnShareAmount;    // Share amount to burn
    uint256 currentDepositEpoch;       // Pool Current Deposit Epoch
    uint256 currentWithdrawEpoch; // Pool Current Withdraw Epoch
}

struct VaultInfo {
    // Info on each pool
    // We Split Pool Struct into 2 struct cause of solidity deep
    bool paused;    // Flag that deposit is paused or not
    uint256 currentDeposit; // Pool Current Pending Deposit Amount
    uint256 currentWithdraw;    // Pool Current Pending Withdraw Amount
    uint256 lockPeriod;          // Pool Guarantee Token Lock Period
    uint256 maxDeposit;         // Pool Max Deposit Amount
    uint256 maxUser;            // Pool Max User Count
    uint256 collectFeeDuration; // Pool Collecting Duration for Fee
    uint256 depositDuration;    // Pool Deposit pending fund to strategy duration
    uint256 withdrawDuration;   // Pool Withdraw fund to user duration
    uint256 lastCollectFee;     // Pool last collect fee timestamp
    uint256 lastDeposit;        // Pool Last deposit pending fund to strategy timestamp
    uint256 lastWithdraw;       // Pool Last Withdraw fund to users timestamp
    uint256 curDepositUser;     // Pool current epoch deposit users length
    uint256 curWithdrawUser;    // Pool current epoch withdraw users length
}

interface ICivVault {

    function getPoolInfo(uint) external view returns(PoolInfo memory, VaultInfo memory);
}

interface ICivFundRT is IERC20, IAccessControl {
    function mint(address to, uint256 amount) external;

    function burn(uint256 amount) external;

    function burnFrom(address account, uint256 amount) external;
}

interface ICivVaultGetter {
    function addUniPair(uint,address,address) external;
    function getPrice(uint,uint) external view returns(uint);
    function getReversePrice(uint,uint) external view returns(uint);
    function getBalanceOfUser(uint,address) external view returns(uint,uint);
    function updateAll(uint) external;
}

interface IERC20Extended is IERC20 {
    function decimals() external view returns (uint256);
    function symbol() external view returns (string memory);
}

// File: contracts/libraries/FixedPoint.sol



// SPDX-License-Identifier: GPL-3.0-or-later
pragma solidity >=0.4.0;

import "./FullMath.sol";

// a library for handling binary fixed point numbers (https://en.wikipedia.org/wiki/Q_(number_format))
library FixedPoint {
    // range: [0, 2**112 - 1]
    // resolution: 1 / 2**112
    struct uq112x112 {
        uint224 _x;
    }

    // range: [0, 2**144 - 1]
    // resolution: 1 / 2**112
    struct uq144x112 {
        uint256 _x;
    }

    uint8 public constant RESOLUTION = 112;
    uint256 public constant Q112 = 0x10000000000000000000000000000; // 2**112
    uint256 private constant Q224 = 0x100000000000000000000000000000000000000000000000000000000; // 2**224
    uint256 private constant LOWER_MASK = 0xffffffffffffffffffffffffffff; // decimal of UQ*x112 (lower 112 bits)

    // encode a uint112 as a UQ112x112
    function encode(uint112 x) internal pure returns (uq112x112 memory) {
        return uq112x112(uint224(x) << RESOLUTION);
    }

    // encodes a uint144 as a UQ144x112
    function encode144(uint144 x) internal pure returns (uq144x112 memory) {
        return uq144x112(uint256(x) << RESOLUTION);
    }

    // decode a UQ112x112 into a uint112 by truncating after the radix point
    function decode(uq112x112 memory self) internal pure returns (uint112) {
        return uint112(self._x >> RESOLUTION);
    }

    // decode a UQ144x112 into a uint144 by truncating after the radix point
    function decode144(uq144x112 memory self) internal pure returns (uint144) {
        return uint144(self._x >> RESOLUTION);
    }

    // multiply a UQ112x112 by a uint, returning a UQ144x112
    // reverts on overflow
    function mul(uq112x112 memory self, uint256 y) internal pure returns (uq144x112 memory) {
        uint256 z = 0;
        require(y == 0 || (z = self._x * y) / y == self._x, 'FixedPoint::mul: overflow');
        return uq144x112(z);
    }

    // multiply a UQ112x112 by an int and decode, returning an int
    // reverts on overflow
    function muli(uq112x112 memory self, int256 y) internal pure returns (int256) {
        uint256 z = FullMath.mulDiv(self._x, uint256(y < 0 ? -y : y), Q112);
        require(z < 2**255, 'FixedPoint::muli: overflow');
        return y < 0 ? -int256(z) : int256(z);
    }

    // multiply a UQ112x112 by a UQ112x112, returning a UQ112x112
    // lossy
    function muluq(uq112x112 memory self, uq112x112 memory other) internal pure returns (uq112x112 memory) {
        if (self._x == 0 || other._x == 0) {
            return uq112x112(0);
        }
        uint112 upper_self = uint112(self._x >> RESOLUTION); // * 2^0
        uint112 lower_self = uint112(self._x & LOWER_MASK); // * 2^-112
        uint112 upper_other = uint112(other._x >> RESOLUTION); // * 2^0
        uint112 lower_other = uint112(other._x & LOWER_MASK); // * 2^-112

        // partial products
        uint224 upper = uint224(upper_self) * upper_other; // * 2^0
        uint224 lower = uint224(lower_self) * lower_other; // * 2^-224
        uint224 uppers_lowero = uint224(upper_self) * lower_other; // * 2^-112
        uint224 uppero_lowers = uint224(upper_other) * lower_self; // * 2^-112

        // so the bit shift does not overflow
        require(upper <= type(uint112).max, 'FixedPoint::muluq: upper overflow');

        // this cannot exceed 256 bits, all values are 224 bits
        uint256 sum = uint256(upper << RESOLUTION) + uppers_lowero + uppero_lowers + (lower >> RESOLUTION);

        // so the cast does not overflow
        require(sum <= type(uint224).max, 'FixedPoint::muluq: sum overflow');

        return uq112x112(uint224(sum));
    }

    // returns a UQ112x112 which represents the ratio of the numerator to the denominator
    // can be lossy
    function fraction(uint256 numerator, uint256 denominator) internal pure returns (uq112x112 memory) {
        require(denominator > 0, 'FixedPoint::fraction: division by zero');
        if (numerator == 0) return FixedPoint.uq112x112(0);

        if (numerator <= type(uint144).max) {
            uint256 result = (numerator << RESOLUTION) / denominator;
            require(result <= type(uint224).max, 'FixedPoint::fraction: overflow');
            return uq112x112(uint224(result));
        } else {
            uint256 result = FullMath.mulDiv(numerator, Q112, denominator);
            require(result <= type(uint224).max, 'FixedPoint::fraction: overflow');
            return uq112x112(uint224(result));
        }
    }

    // take the reciprocal of a UQ112x112
    // reverts on overflow
    // lossy
    function reciprocal(uq112x112 memory self) internal pure returns (uq112x112 memory) {
        require(self._x != 0, 'FixedPoint::reciprocal: reciprocal of zero');
        require(self._x != 1, 'FixedPoint::reciprocal: overflow');
        return uq112x112(uint224(Q224 / self._x));
    }
}

// File: contracts/libraries/UniswapV2OracleLibrary.sol



pragma solidity >=0.5.0;

import '@uniswap/v2-core/contracts/interfaces/IUniswapV2Pair.sol';
import './FixedPoint.sol';

// library with helper methods for oracles that are concerned with computing average prices
library UniswapV2OracleLibrary {
    using FixedPoint for *;

    // helper function that returns the current block timestamp within the range of uint32, i.e. [0, 2**32 - 1]
    function currentBlockTimestamp() internal view returns (uint32) {
        return uint32(block.timestamp % 2 ** 32);
    }

    // produces the cumulative price using counterfactuals to save gas and avoid a call to sync.
    function currentCumulativePrices(
        address pair
    ) internal view returns (uint price0Cumulative, uint price1Cumulative, uint32 blockTimestamp) {
        blockTimestamp = currentBlockTimestamp();
        price0Cumulative = IUniswapV2Pair(pair).price0CumulativeLast();
        price1Cumulative = IUniswapV2Pair(pair).price1CumulativeLast();

        // if time has elapsed since the last update on the pair, mock the accumulated price values
        (uint112 reserve0, uint112 reserve1, uint32 blockTimestampLast) = IUniswapV2Pair(pair).getReserves();
        if (blockTimestampLast != blockTimestamp) {
            // subtraction overflow is desired
            uint32 timeElapsed = blockTimestamp - blockTimestampLast;
            // addition overflow is desired
            // counterfactual
            price0Cumulative += uint(FixedPoint.fraction(reserve1, reserve0)._x) * timeElapsed;
            // counterfactual
            price1Cumulative += uint(FixedPoint.fraction(reserve0, reserve1)._x) * timeElapsed;
        }
    }
}
// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `sender` to `recipient`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
        }
        _balances[to] += amount;

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        _balances[account] += amount;
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
        }
        _totalSupply -= amount;

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/access/IAccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router01.sol



pragma solidity >=0.6.2;

interface IUniswapV2Router01 {
    function factory() external pure returns (address);
    function WETH() external pure returns (address);

    function addLiquidity(
        address tokenA,
        address tokenB,
        uint amountADesired,
        uint amountBDesired,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB, uint liquidity);
    function addLiquidityETH(
        address token,
        uint amountTokenDesired,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external payable returns (uint amountToken, uint amountETH, uint liquidity);
    function removeLiquidity(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETH(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountToken, uint amountETH);
    function removeLiquidityWithPermit(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETHWithPermit(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountToken, uint amountETH);
    function swapExactTokensForTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapTokensForExactTokens(
        uint amountOut,
        uint amountInMax,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapExactETHForTokens(uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);
    function swapTokensForExactETH(uint amountOut, uint amountInMax, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapExactTokensForETH(uint amountIn, uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapETHForExactTokens(uint amountOut, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);

    function quote(uint amountA, uint reserveA, uint reserveB) external pure returns (uint amountB);
    function getAmountOut(uint amountIn, uint reserveIn, uint reserveOut) external pure returns (uint amountOut);
    function getAmountIn(uint amountOut, uint reserveIn, uint reserveOut) external pure returns (uint amountIn);
    function getAmountsOut(uint amountIn, address[] calldata path) external view returns (uint[] memory amounts);
    function getAmountsIn(uint amountOut, address[] calldata path) external view returns (uint[] memory amounts);
}

// File: contracts/libraries/FullMath.sol



// SPDX-License-Identifier: MIT
pragma solidity >=0.4.0;

/// @title Contains 512-bit math functions
/// @notice Facilitates multiplication and division that can have overflow of an intermediate value without any loss of precision
/// @dev Handles "phantom overflow" i.e., allows multiplication and division where an intermediate value overflows 256 bits
library FullMath {
    /// @notice Calculates floor(a×b÷denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
    /// @param a The multiplicand
    /// @param b The multiplier
    /// @param denominator The divisor
    /// @return result The 256-bit result
    /// @dev Credit to Remco Bloemen under MIT license https://xn--2-umb.com/21/muldiv
    function mulDiv(
        uint256 a,
        uint256 b,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        // 512-bit multiply [prod1 prod0] = a * b
        // Compute the product mod 2**256 and mod 2**256 - 1
        // then use the Chinese Remainder Theorem to reconstruct
        // the 512 bit result. The result is stored in two 256
        // variables such that product = prod1 * 2**256 + prod0
        uint256 prod0; // Least significant 256 bits of the product
        uint256 prod1; // Most significant 256 bits of the product
        assembly {
            let mm := mulmod(a, b, not(0))
            prod0 := mul(a, b)
            prod1 := sub(sub(mm, prod0), lt(mm, prod0))
        }

        // Handle non-overflow cases, 256 by 256 division
        if (prod1 == 0) {
            require(denominator > 0);
            assembly {
                result := div(prod0, denominator)
            }
            return result;
        }

        // Make sure the result is less than 2**256.
        // Also prevents denominator == 0
        require(denominator > prod1);

        ///////////////////////////////////////////////
        // 512 by 256 division.
        ///////////////////////////////////////////////

        // Make division exact by subtracting the remainder from [prod1 prod0]
        // Compute remainder using mulmod
        uint256 remainder;
        assembly {
            remainder := mulmod(a, b, denominator)
        }
        // Subtract 256 bit number from 512 bit number
        assembly {
            prod1 := sub(prod1, gt(remainder, prod0))
            prod0 := sub(prod0, remainder)
        }

        // Factor powers of two out of denominator
        // Compute largest power of two divisor of denominator.
        // Always >= 1.
        //uint256 twos = -denominator & denominator;
        uint256 twos = denominator & (~denominator + 1);
        // Divide denominator by power of two
        assembly {
            denominator := div(denominator, twos)
        }

        // Divide [prod1 prod0] by the factors of two
        assembly {
            prod0 := div(prod0, twos)
        }
        // Shift in bits from prod1 into prod0. For this we need
        // to flip `twos` such that it is 2**256 / twos.
        // If twos is zero, then it becomes one
        assembly {
            twos := add(div(sub(0, twos), twos), 1)
        }
        prod0 |= prod1 * twos;

        // Invert denominator mod 2**256
        // Now that denominator is an odd number, it has an inverse
        // modulo 2**256 such that denominator * inv = 1 mod 2**256.
        // Compute the inverse by starting with a seed that is correct
        // correct for four bits. That is, denominator * inv = 1 mod 2**4
        uint256 inv = (3 * denominator) ^ 2;
        // Now use Newton-Raphson iteration to improve the precision.
        // Thanks to Hensel's lifting lemma, this also works in modular
        // arithmetic, doubling the correct bits in each step.
        inv *= 2 - denominator * inv; // inverse mod 2**8
        inv *= 2 - denominator * inv; // inverse mod 2**16
        inv *= 2 - denominator * inv; // inverse mod 2**32
        inv *= 2 - denominator * inv; // inverse mod 2**64
        inv *= 2 - denominator * inv; // inverse mod 2**128
        inv *= 2 - denominator * inv; // inverse mod 2**256

        // Because the division is now exact we can divide by multiplying
        // with the modular inverse of denominator. This will give us the
        // correct result modulo 2**256. Since the precoditions guarantee
        // that the outcome is less than 2**256, this is the final result.
        // We don't need to compute the high bits of the result and prod1
        // is no longer required.
        result = prod0 * inv;
        return result;
    }

    /// @notice Calculates ceil(a×b÷denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
    /// @param a The multiplicand
    /// @param b The multiplier
    /// @param denominator The divisor
    /// @return result The 256-bit result
    function mulDivRoundingUp(
        uint256 a,
        uint256 b,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        result = mulDiv(a, b, denominator);
        if (mulmod(a, b, denominator) > 0) {
            require(result < type(uint256).max);
            result++;
        }
    }
}